MODULE *Sailfish*

Specification of the signature-free *Sailfish* consensus algorithm at a high level of abstraction.

We use a number of abstractions and simplifiying assumptions in order to expose the high-level principles of the algorithm clearly and in order to make model-checking of interesting configuations tractable:

1) Nodes read and write a global DAG. When a node transitions into a new round, it is provided with an arbitrary quorum of vertices from the previous round (except that, after GST, some additional assumptions apply).

2) We do not model timeouts. Instead, we assume that, every round r after GST, each correct node votes for the previous leader.

3) Byzantine nodes are allowed to create new DAG vertices arbitrarily, but only one per round.

4) We do not explicitly model committing based on $2f + 1$ first RBC messages.

5) There are no weak edges.

EXTENDS *Digraph*, *Integers*, *FiniteSets*, *Sequences*

CONSTANTS

N The set of all nodes
, F The set of *Byzantine* nodes
, R The set of rounds
, $IsQuorum(\_)$ Whether a set is a quorum (*i.e.* cardinality $\geq n$-f)
, $IsBlocking(\_)$ Whether a set is a blocking set (*i.e.* cardinality $\geq f + 1$)
, $Leader(\_)$ operator mapping each round to its leader
, GST the first round in which the system is synchronous

ASSUME $\exists\, n \in R : R = 1 \,..\, n$ useful rounds start at 1

For our purpose of checking safety and liveness of *Sailfish*, we do not need to model blocks of transactions. Instead, DAG vertices just consist of a node and a round.

$V \triangleq N \times R$ the set of possible DAG vertices
$Node(v) \triangleq v[1]$
$Round(v) \triangleq v[2]$

Next we define how we order DAG vertices when we commit a leader vertice

$LeaderVertice(r) \triangleq \langle Leader(r),\, r \rangle$

RECURSIVE $OrderSet(\_)$ arbitrarily order the members of the set S
$OrderSet(S) \triangleq$ IF $S = \{\}$ THEN $\langle\rangle$ ELSE
LET $e \triangleq$ CHOOSE $e \in S$: TRUE
IN $Append(OrderSet(S \setminus \{e\}),\, e)$

NOTE: CHOOSE is deterministic in TLA+,
i.e. CHOOSE $e \in S$: TRUE is always the same e if S is the same

```
RECURSIVE CollectLeaders(_, _, _)
CollectLeaders(vs, r, dag) ≜ IF vs = {} THEN ⟨⟩ ELSE
    LET children ≜ UNION {Children(v, dag) : v ∈ vs}
    IN  IF LeaderVertice(r) ∈ vs
        THEN Append(
                CollectLeaders(Children(LeaderVertice(r), dag), r − 1, dag),
                LeaderVertice(r))
        ELSE CollectLeaders(children, r − 1, dag)

RECURSIVE OrderVertices(_, _)
OrderVertices(dag, leaderVertices) ≜
    IF leaderVertices = ⟨⟩ THEN ⟨⟩ ELSE
    LET l ≜ Head(leaderVertices)
        toOrder ≜ Descendants({l}, dag)
        prefix ≜ OrderSet(toOrder)
        remainingVertices ≜ Vertices(dag) \ (toOrder ∪ {l})
        remainingEdges ≜ {e ∈ Edges(dag) : {e[1], e[2]} ⊆ remainingVertices}
        remainingDAG ≜ ⟨remainingVertices, remainingEdges⟩
    IN  prefix ∘ ⟨l⟩ ∘ OrderVertices(remainingDAG, Tail(leaderVertices))

CommitLeader(v, dag) ≜
    LET leaderVertices ≜ CollectLeaders({v}, Round(v), dag)
    IN  OrderVertices(dag, leaderVertices)
```

Now we specify the algorithm in the *PlusCal* language.

```
--algorithm Sailfish{
  variables
    vs = {},    the vertices of the DAG
    es = {};    the edges of the DAG
  define {
    dag ≜ ⟨vs, es⟩
    NoVoteQuorum(r, delivered) ≜
      LET NoVote ≜ {v ∈ delivered : LeaderVertice(r − 1) ∉ Children(v, dag)}
      IN  IsQuorum({Node(v) : v ∈ NoVote})
  }
  process ( correctNode ∈ N \ F )
    variables
      round = 0,    current round; 0 means the node has not started execution
      log = ⟨⟩ ;    delivered log
  {
l0:   while ( TRUE ) {
        if ( round = 0 ) {    start the first round r = 1
          round := 1 ;
          vs := vs ∪ {⟨self, 1⟩}
        }
```

```
            else {   start a round r > 1
                with ( r ∈ {r ∈ R : r > round} )
                with ( delivered ∈ SUBSET {v ∈ vs : Round(v) = r − 1} ) {
                    await IsQuorum({Node(v) : v ∈ delivered}) ;
                    await LeaderVertice(r − 1) ∈ delivered ⇒
                              ∨ LeaderVertice(r − 2) ∈ Children(LeaderVertice(r − 1), dag)
                              ∨ NoVoteQuorum(r − 1, delivered) ;
                    if ( Leader(r) = self )
                        await    ∨ LeaderVertice(r − 1) ∈ delivered
                                 ∨ NoVoteQuorum(r, delivered) ;
                    round := r ;
                    with ( newV = ⟨self, round⟩ ) {
                        vs := vs ∪ {newV} ;
                        es := es ∪ {⟨newV, pv⟩ : pv ∈ delivered} ;
                    } ;
                     commit if there is a quorum of votes for the leader of r − 2:
                    if ( round > 1 )
                        with ( votesForLeader = {pv ∈ delivered : ⟨pv, LeaderVertice(round − 2)⟩ ∈ es} )
                        if ( IsBlocking({Node(pv) : pv ∈ votesForLeader}) )
                            log := CommitLeader(LeaderVertice(round − 2), dag)
                }
            }
        }
    }
```

Next comes our model of *Byzantine* nodes. Because the real protocol disseminates *DAG* vertices using reliable broadcast, *Byzantine* nodes cannot equivocate and cannot deviate much from the protocol (lest their messages be ignored).

```
    process ( byzantineNode ∈ F )
    {
l0:     while ( TRUE ) {
            with ( r ∈ R )
            with ( newV = ⟨self, r⟩ ) {
                when newV ∉ vs ;   no equivocation
                if ( r = 1 )
                    vs := vs ∪ {newV}
                else
                with ( delivered ∈ SUBSET {v ∈ vs : Round(v) = r − 1} ) {
                    await IsQuorum({Node(v) : v ∈ delivered}) ;   ignored otherwise
                    vs := vs ∪ {newV} ;
                    es := es ∪ {⟨newV, pv⟩ : pv ∈ delivered}
                }
            }
        }
    }
}
```

Next we define the safety and liveness properties

$Compatible(s1,\ s2) \triangleq$ whether the sequence $s1$ is a prefix of the sequence $s2$, or vice versa
LET $Min(n1,\ n2) \triangleq$ IF $n1 \geq n2$ THEN $n2$ ELSE $n1$IN
$\forall\, i \in 1\,..\,Min(Len(s1),\ Len(s2)) : s1[i] = s2[i]$

$Agreement \triangleq \forall\, n1,\ n2 \in N \setminus F : Compatible(log[n1],\ log[n2])$

$Liveness \triangleq \forall\, r \in R : r \geq GST \wedge Leader(r) \notin F \Rightarrow \exists\, B \in$ SUBSET $(N \setminus F) :$
$\wedge$ $IsBlocking(B)$
$\wedge \forall\, n \in B : round[n] \geq r + 2 \Rightarrow \exists\, i \in$ DOMAIN $log[n] : log[n][i] = LeaderVertice(r)$

Finally we make a few auxiliary definitions used for model-checking with TLC

Basic typing invariant:
$TypeOK \triangleq$
$\wedge$ $\forall\, v \in vs : Node(v) \in N \wedge Round(v) \in Nat \setminus \{0\}$
$\wedge$ $\forall\, e \in es :$
$\wedge$ $e = \langle e[1],\ e[2] \rangle$
$\wedge$ $\{e[1],\ e[2]\} \subseteq vs$
$\wedge$ $Round(e[1]) > Round(e[2])$
$\wedge$ $\forall\, n \in N \setminus F : round[n] \in Nat$

Synchrony assumption: for each round r from GST onwards, if the leader of r is correct then every correct node votes for the round-r leader vertex in round
$r + 1$
$Synchrony \triangleq \forall\, r \in R : r \geq GST \wedge Leader(r) \notin F \Rightarrow \forall\, n \in N \setminus F :$
LET $v \triangleq \langle n,\ r + 1 \rangle$
IN $v \in vs \Rightarrow LeaderVertice(r) \in Children(v,\ dag)$

We add the synchrony assumption to the specification
$SyncNext \triangleq (\exists\, self \in N \setminus F : correctNode(self) \wedge Synchrony')$
$\vee\, (\exists\, self \in F : byzantineNode(self))$
$SyncSpec \triangleq Init \wedge \Box[SyncNext]_{vars}$

Next we define a constraint to stop the model-checker.
$Max(S) \triangleq$ CHOOSE $x \in S : \forall\, y \in S : y \leq x$
$StateConstraint \triangleq \forall\, n \in N \setminus F : round[n] \in 0\,..\,Max(R)$

Finally, we give some properties we expect to be violated (useful to get the model-checker to print interesting executions).

$Falsy1 \triangleq \neg($
$\forall\, n \in N \setminus F : round[n] = Max(R)$
$)$

$Falsy2 \triangleq \neg($

$$\exists\, n \in N \setminus F : Len(log[n]) > 1$$
)